

G11SAF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G11SAF fits a latent variable model (with a single factor) to data consisting of a set of measurements on individuals in the form of binary-valued sequences (generally referred to as score patterns). Various measures of goodness of fit are calculated along with the factor (theta) scores.

2 Specification

```

SUBROUTINE G11SAF(IP, N, GPROB, IS, X, NRX, IRL, A, C, IPRINT,
1          CGETOL, MAXIT, CHISQR, ISHOW, NITER, ALPHA,
2          PIGAM, CM, ICM, G, EXPP, NRE, OBS, EXF, Y, XL,
3          IOB, RLOGL, CHI, IDF, SIGLEV, W, LW, IFAIL)
  INTEGER
1    IP, N, IS, NRX, IRL(IS), IPRINT, MAXIT, ISHOW,
  NITER, ICM, NRE, IOB(IS), IDF, LW, IFAIL
  real
1    A(IP), C(IP), CGETOL, ALPHA(IP), PIGAM(IP),
2    CM(ICM,2*IP), G(2*IP), EXPP(NRE,IP),
3    OBS(NRE,IP), EXF(IS), Y(IS), XL(IS), RLOGL, CHI,
  SIGLEV, W(LW)
  LOGICAL
1    GPROB, X(NRX,IP), CHISQR

```

3 Description

Given a set of p dichotomous variables $\underline{x} = (x_1, x_2, \dots, x_p)'$, where $'$ denotes vector or matrix transpose, the objective is to investigate whether the association between them can be adequately explained by a latent variable model of the form (see Bartholomew [1] and [2]):

$$G\{\pi_i(\theta)\} = \alpha_{i0} + \alpha_{i1}\theta \quad (1)$$

The x_i are called item responses and take the value 0 or 1. θ denotes the latent variable assumed to have a standard Normal distribution over a population of individuals to be tested on p items. Call $\pi_i(\theta) = P(x_i = 1|\theta)$ the item response function: it represents the probability that an individual with latent ability θ will produce a positive response (1) to item i . α_{i0} and α_{i1} are item parameters which can assume any real values. The set of parameters $\{\alpha_{i1}, \text{ for } i = 1, 2, \dots, p\}$ being coefficients of the unobserved variable θ , can be interpreted as 'factor loadings'.

G is a function selected by the user as either Φ^{-1} or logit, mapping the interval (0,1) onto the whole real line. Data from a random sample of n individuals takes the form of the matrices X and R defined below:

$$X_{s \times p} = \begin{bmatrix} x_{11} & x_{12} & \cdot & \cdot & \cdot & x_{1p} \\ x_{21} & x_{22} & \cdot & \cdot & \cdot & x_{2p} \\ \cdot & \cdot & & & & \cdot \\ \cdot & \cdot & & & & \cdot \\ \cdot & \cdot & & & & \cdot \\ x_{s1} & x_{s2} & \cdot & \cdot & \cdot & x_{sp} \end{bmatrix} = \begin{bmatrix} \underline{x}'_1 \\ \underline{x}'_2 \\ \cdot \\ \cdot \\ \cdot \\ \underline{x}'_s \end{bmatrix}, \quad R_{s \times 1} = \begin{bmatrix} r_1 \\ r_2 \\ \cdot \\ \cdot \\ \cdot \\ r_s \end{bmatrix}$$

where $\underline{x}_l = (x_{l1}, x_{l2}, \dots, x_{lp})'$ denotes the l th score pattern in the sample, r_l the frequency with which \underline{x}_l occurs and s the number of different score patterns observed. (Thus $\sum_{l=1}^s r_l = n$). It can be shown that the log likelihood function is proportional to

$$\sum_{l=1}^s r_l \log P_l$$

where

$$P_l = P(\underline{x} = \underline{x}_l) = \int_{-\infty}^{\infty} P(\underline{x} = \underline{x}_l | \theta) \phi(\theta) d\theta \quad (2)$$

($\phi(\theta)$ being the probability density function of a standard Normal random variable).

P_l denotes the unconditional probability of observing score pattern \underline{x}_l . The integral in (2) is approximated using Gauss–Hermite quadrature. If we take $G(z) = \text{logit } z = \log\left(\frac{z}{1-z}\right)$ in (1) and reparametrise as follows:

$$\begin{aligned} \alpha_i &= \alpha_{i1}, \\ p_i &= \text{logit}^{-1} \alpha_{i0} \end{aligned}$$

(1) reduces to the logit model (see Bartholomew [1])

$$\pi_i(\theta) = \frac{\pi_i}{\pi_i + (1 - \pi_i) \exp(-\alpha_i \theta)}.$$

If we take $G(z) = \Phi^{-1}(z)$ (where Φ is the cumulative distribution function of a standard Normal random variable) and reparametrise as follows

$$\begin{aligned} \alpha_i &= \frac{\alpha_{i1}}{\sqrt{(1 + \alpha_{i1}^2)}} \\ \gamma_i &= \frac{-\alpha_{i0}}{\sqrt{(1 + \alpha_{i1}^2)}}. \end{aligned}$$

(1) reduces to the probit model (see Bock and Aitkin [3])

$$\pi_i(\theta) = \phi\left(\frac{\alpha_i \theta - \gamma_i}{\sqrt{(1 - \alpha_i^2)}}\right).$$

An E-M algorithm (see Bock and Aitkin [3]) is used to maximize the log likelihood function. The number of quadrature points used is set initially to 10 and once convergence is attained increased to 20.

The theta score of an individual responding in score pattern \underline{x}_l is computed as the posterior mean, i.e., $E(\theta | \underline{x}_l)$. For the logit model the component score $X_l = \sum_{j=1}^p \alpha_j x_{lj}$ is also calculated. (Note that in calculating the theta scores and measures of goodness of fit G11SAF automatically reverses the coding on item j if $\alpha_j < 0$ – it is assumed in the model that a response at the one level is showing a higher measure of latent ability than a response at the zero level.)

The frequency distribution of score patterns is required as input data. If the user's data is in the form of individual score patterns (uncounted), then G11SBF may be used to calculate the frequency distribution.

4 References

- [1] Bartholomew D J (1980) Factor analysis for categorical data (with Discussion) *J. Roy. Statist. Soc. Ser. B* **42** 293–321
- [2] Bartholomew D J (1987) *Latent Variable Models and Factor Analysis* Griffin
- [3] Bock R D and Aitkin M (1981) Marginal maximum likelihood estimation of item parameters: Application of an E–M algorithm *Psychometrika* **46** 443–459

5 Parameters

- 1: IP — INTEGER *Input*
On entry: the number of dichotomous variables, p .
Constraint: IP \geq 3.
- 2: N — INTEGER *Input*
On entry: the number of individuals in the sample, n .
Constraint: N \geq 7.

- 3:** GPROB — LOGICAL *Input*
On entry: GPROB must be set equal to .TRUE. if $G(z) = \Phi^{-1}(z)$ and .FALSE. if $G(z) = \text{logit } z$.
- 4:** IS — INTEGER *Input*
On entry: IS must be set equal to the number of different score patterns in the sample, s .
Constraint: $2 \times \text{IP} < \text{IS} \leq \min(2^{\text{IP}}, N)$.
- 5:** X(NRX,IP) — LOGICAL array *Input/Output*
On entry: the first s rows of X must contain the s different score patterns. The l th row of X must contain the l th score pattern with $X(l, j)$ set equal to .TRUE. if $x_{lj} = 1$ and .FALSE. if $x_{lj} = 0$. All rows of X must be distinct.
On exit: if IFAIL = 0 or IFAIL \geq 3 then the first s rows of X still contain the s different score patterns. However, the following points should be noted:
- (1) If the estimated factor loading for the j th item is negative then that item is re-coded, i.e., 0's and 1's (or .TRUE. and .FALSE.) in the j th column of X are interchanged.
 - (2) The rows of X will be re-ordered so that the theta scores corresponding to rows of X are in increasing order of magnitude.
- 6:** NRX — INTEGER *Input*
On entry: the first dimension of the array X as declared in the (sub)program from which G11SAF is called.
Constraint: $\text{NRX} \geq \text{IS}$.
- 7:** IRL(IS) — INTEGER array *Input/Output*
On entry: the i th component of IRL must be set equal to the frequency with which the i th row of X occurs.
Constraints:
- $$\text{IRL}(i) \geq 0 \text{ for } i = 1, 2, \dots, s.$$
- $$\sum_{i=1}^s \text{IRL}(i) = n.$$
- On exit:* if IFAIL = 0 or IFAIL \geq 3 then the first s components of IRL are re-ordered as are the rows of X.
- 8:** A(IP) — *real* array *Input/Output*
On entry: $A(j)$ must be set equal to an initial estimate of α_{j1} . **In order to avoid divergence problems with the E-M algorithm the user is strongly advised to set all the $A(j)$ to 0.5.**
On exit: $A(j)$ contains the latest estimate of α_{j1} , for $j = 1, 2, \dots, p$. (Because of possible recoding all elements of A will be positive.)
- 9:** C(IP) — *real* array *Input/Output*
On entry: $C(j)$ must be set equal to an initial estimate of α_{j0} . **In order to avoid divergence problems with the E-M algorithm the user is strongly advised to set all the $C(j)$'s to 0.0.**
On exit: $C(j)$ contains the latest estimate of α_{j0} , for $j = 1, 2, \dots, p$.
- 10:** IPRINT — INTEGER *Input*
On entry: the frequency with which the maximum likelihood search routine is to be monitored.
 If IPRINT $>$ 0, the search is monitored once every IPRINT iterations, and when the number of quadrature points is increased, and again at the final solution point.
 If IPRINT = 0, the search is monitored once at the final point.
 If IPRINT $<$ 0, the search is not monitored at all.
 IPRINT should normally be set to a small positive number.
Suggested value: IPRINT = 1.

- 11: CGETOL** — *real* *Input*
On entry: the accuracy to which the solution is required. If CGETOL is set to 10^{-l} and on exit IFAIL = 0 or 7, then all elements of the gradient vector will be smaller than 10^{-l} in absolute value. For most practical purposes the value 10^{-4} should suffice. The user should be wary of setting CGETOL too small since the convergence criterion may then have become too strict for the machine to handle. If CGETOL has been set to a value which is less than the square root of the *machine precision*, ϵ , then G11SAF will use the value $\sqrt{\epsilon}$ instead.
- 12: MAXIT** — INTEGER *Input*
On entry: the maximum number of iterations to be made in the maximum likelihood search. There will be an error exit (see Section 6) if the search routine has not converged in MAXIT iterations.
Constraint: MAXIT ≥ 1 .
Suggested value: MAXIT = 1000.
- 13: CHISQR** — LOGICAL *Input*
On entry: if CHISQR is set equal to .TRUE., then a likelihood ratio statistic will be calculated (see CHI).
 If CHISQR is set equal to .FALSE., no such statistic will be calculated.
- 14: ISHOW** — INTEGER *Input*
On entry: indicates which of the following three quantities are to be printed before exit from the routine (except when IFAIL = 1 or 2):
- (a) Table of maximum likelihood estimates and standard errors (as returned in the output arrays A, C, ALPHA, PIGAM and CM).
 - (b) Table of observed and expected first and second order margins (as returned in the output arrays EXPP and OBS).
 - (c) Table of observed and expected frequencies of score patterns along with theta scores (as returned in the output arrays IRL, EXF, Y, XL and IOB) and the likelihood ratio statistic (if required).
- ISHOW = 0 none of the above are printed
 ISHOW = 1 (a) only is printed
 ISHOW = 2 (b) only is printed
 ISHOW = 3 (c) only is printed
 ISHOW = 4 (a) and (b) are printed
 ISHOW = 5 (a) and (c) are printed
 ISHOW = 6 (b) and (c) are printed
 ISHOW = 7 (a), (b) and (c) are printed
- Constraint:* $0 \leq \text{ISHOW} \leq 7$.
- 15: NITER** — INTEGER *Output*
On exit: if IFAIL = 0 or IFAIL ≥ 3 then NITER contains the number of iterations performed by the maximum likelihood search routine.
- 16: ALPHA(IP)** — *real* array *Output*
On exit: if IFAIL = 0 or IFAIL ≥ 3 then ALPHA(*j*) contains the latest estimate of α_j . (Because of possible re-coding all elements of ALPHA will be positive.)
- 17: PIGAM(IP)** — *real* array *Output*
On exit: if IFAIL = 0 or IFAIL ≥ 3 then PIGAM(*j*) contains the latest estimate of either π_j if GPROB = .FALSE. (logit model) or γ_j if GPROB = .TRUE. (probit model).

- 18:** CM(ICM,2*IP) — *real* array *Output*
On exit: if IFAIL = 0 or IFAIL ≥ 3 then the strict lower triangle of CM contains the correlation matrix of the parameter estimates held in ALPHA and PIGAM on exit. The diagonal elements of CM contain the standard errors. Thus:
- For $i = 1, 2, \dots, p$
- | | | |
|--|---|--|
| CM($2 \times i - 1, 2 \times i - 1$) | = | standard error (ALPHA(i)) |
| CM($2 \times i, 2 \times i$) | = | standard error (PIGAM(i)) |
| CM($2 \times i, 2 \times i - 1$) | = | correlation (PIGAM(i), ALPHA(i)) |
- For $j = 1, 2, \dots, i - 1$
- | | | |
|--|---|--|
| CM($2 \times i - 1, 2 \times j - 1$) | = | correlation (ALPHA(i), ALPHA(j)) |
| CM($2 \times i, 2 \times j$) | = | correlation (PIGAM(i), PIGAM(j)) |
| CM($2 \times i - 1, 2 \times j$) | = | correlation (ALPHA(i), PIGAM(j)) |
| CM($2 \times i, 2 \times j - 1$) | = | correlation (ALPHA(j), PIGAM(i)) |
- If the second derivative matrix cannot be computed then all the elements of CM are returned as zero.
- 19:** ICM — INTEGER *Input*
On entry: the first dimension of the array CM as declared in the (sub)program from which G11SAF is called.
Constraint: ICM ≥ IP + IP.
- 20:** G(2*IP) — *real* array *Output*
On exit: if IFAIL = 0 or IFAIL ≥ 3 then G contains the estimated gradient vector corresponding to the final point held in the arrays ALPHA and PIGAM. G($2 \times j - 1$) contains the derivative of the log likelihood with respect to ALPHA(j) for $j = 1, 2, \dots, p$. G($2 \times j$) contains the derivative of the log likelihood with respect to PIGAM(j) for $j = 1, 2, \dots, p$.
- 21:** EXPP(NRE,IP) — *real* array *Output*
On exit: if IFAIL = 0 or IFAIL ≥ 3 then EXPP(i, j) contains the expected percentage of individuals in the sample who respond positively to items i and j , corresponding to the final point held in the arrays ALPHA and PIGAM.
- 22:** NRE — INTEGER *Input*
On entry: the first dimension of the arrays EXPP and OBS as declared in the (sub)program from which G11SAF is called.
Constraint: NRE ≥ IP.
- 23:** OBS(NRE,IP) — *real* array *Output*
On exit: if IFAIL = 0 or IFAIL ≥ 3 then OBS(i, j) contains the observed percentage of individuals in the sample who responded positively to items i and j .
- 24:** EXF(IS) — *real* array *Output*
On exit: if IFAIL = 0 or IFAIL ≥ 3 then EXF(l) contains the expected frequency of the l th score pattern (l th row of X), corresponding to the final point held in the arrays ALPHA and PIGAM.
- 25:** Y(IS) — *real* array *Output*
On exit: if IFAIL = 0 or IFAIL ≥ 3 then Y(l) contains the estimated theta score corresponding to the l th row of X, for the final point held in the arrays ALPHA and PIGAM.

26: XL(IS) — *real* array *Workspace*

If GPROB has been set equal to .FALSE. (logit model) then on exit if IFAIL = 0 or IFAIL ≥ 3, XL(*l*) contains the estimated component score corresponding to the *l*th row of X for the final point held in the arrays ALPHA and PIGAM. If GPROB is set equal to .TRUE. (probit model) this array is not used.

27: IOB(IS) — INTEGER array *Output*

On exit: if IFAIL = 0 or IFAIL ≥ 3 then IOB(*l*) contains the number of items in the *l*th row of X for which the response was positive (.TRUE.).

28: RLOGL — *real* *Output*

On exit: if IFAIL = 0 or IFAIL ≥ 3 then RLOGL contains the value of the log likelihood kernel corresponding to the final point held in the arrays ALPHA and PIGAM, viz

$$\sum_{l=1}^s \text{IRL}(l) \times \log(\text{EXF}(l)/n).$$

29: CHI — *real* *Output*

On exit: if CHISQR was set equal to .TRUE. on entry, then if IFAIL = 0 or IFAIL ≥ 3 CHI will contain the value of the likelihood ratio statistic corresponding to the final parameter estimates held in the arrays ALPHA and PIGAM, viz

$$2 \times \sum_{l=1}^s \text{IRL}(l) \times \log(\text{EXF}(l)/\text{IRL}(l)).$$

The summation is over those elements of IRL which are positive. If EXF(*l*) is less than 5.0, then adjacent score patterns are pooled (the score patterns in X being first put in order of increasing theta score).

If CHISQR has been set equal to .FALSE., then CHI is not used.

30: IDF — INTEGER *Output*

On exit: if CHISQR was set equal to .TRUE. on entry, then if IFAIL = 0 or IFAIL ≥ 3 IDF will contain the degrees of freedom associated with the likelihood ratio statistic, CHI.

$$\begin{aligned} \text{IDF} &= s_0 - 2 \times p && \text{if } s_0 < 2^p \\ \text{IDF} &= s_0 - 2 \times p - 1 && \text{if } s_0 = 2^p \end{aligned}$$

where s_0 denotes the number of terms summed to calculate CHI ($s_0 = s$ only if there is no pooling).

If CHISQR has been set equal to .FALSE., then IDF is not used.

31: SIGLEV — *real* *Output*

On exit: if CHISQR was set equal to .TRUE. on entry, then if IFAIL = 0 or IFAIL ≥ 3 SIGLEV will contain the significance level of CHI based on IDF degrees of freedom. If IDF is zero or negative then SIGLEV is set to zero. If CHISQR was set equal to .FALSE., then SIGLEV is not used.

32: W(LW) — *real* array *Workspace*

33: LW — INTEGER *Input*

On entry: the dimension of the array W as declared in the (sub)program from which G11SAF is called.

Constraint: LW ≥ 4 × IP × (IP + 16).

34: IFAIL — INTEGER*Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error or gives a warning (see Section 6).

For this routine, because the values of output parameters may be useful even if IFAIL \neq 0 on exit, users are recommended to set IFAIL to -1 before entry. **It is then essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings specified by the routine:

IFAIL = 1

- On entry, $IP < 3$,
- or $N < 7$,
- or $IS \leq 2 \times IP$,
- or $IS > N$,
- or $IS > 2^{IP}$,
- or two or more rows of X are identical,
- or $NRX < IS$,
- or $\sum_{l=1}^{IS} IRL(l) \neq N$,
- or at least one of $IRL(l) < 0$, for $l = 1, 2, \dots, IS$,
- or $MAXIT < 1$,
- or $ISHOW < 0$,
- or $ISHOW > 7$,
- or $ICM < IP + IP$,
- or $NRE < IP$,
- or $LW < 4 \times IP \times (IP + 16)$.

IFAIL = 2

For at least one of the IP items the responses are all at the same level. To proceed, the user must delete this item from his data set.

IFAIL = 3

There have been MAXIT iterations of the maximum likelihood search routine. If steady increases in the log likelihood kernel were monitored up to the point where this exit occurred, then the exit probably occurred simply because MAXIT was set too small, so the calculations should be restarted from the final point held in A and C. This type of exit may also indicate that there is no maximum to the likelihood surface.

IFAIL = 4

One of the elements of A has exceeded 10.0 in absolute value (see Section 8.3). If steady increases in the log-likelihood kernel were monitored up to the point where this exit occurred then this exit may indicate that there is no maximum to the likelihood surface. The user is advised to restart the calculations from a different point to see whether the E-M algorithm moves off in the same direction.

IFAIL = 5

This indicates a failure in F01ABF which is used to invert the second derivative matrix for calculating the variance-covariance matrix of parameter estimates. It was also found that MAXIT iterations had been performed (see IFAIL = 3).

The elements of CM will then have been set to zero on exit. The user is advised to restart the calculations with a larger value for MAXIT.

IFAIL = 6

This indicates a failure in F01ABF which is used to invert the second derivative matrix for calculating the variance-covariance matrix of parameter estimates. It was also found that one of the elements of A had exceeded 10.0 in absolute value (see IFAIL = 4). The elements of CM will have then been set to zero on exit. The user is advised to restart the calculations from a different point to see whether the E-M algorithm moves off in the same direction.

IFAIL = 7

If CHISQR was set equal to .TRUE. on entry, so that a likelihood ratio statistic was calculated, then IFAIL = 7 merely indicates that the value of IDF on exit is ≤ 0 , i.e., the χ^2 statistic is meaningless. In this case SIGLEV is returned as zero. **All other output information should be correct, i.e., can be treated as if IFAIL were 0 on exit.**

7 Accuracy

On exit from G11SAF if IFAIL = 0 or 7 then the following condition will be satisfied:

$$\text{Max}_{1 \leq i \leq 2 \times p} \{|G(i)|\} < \text{CGETOL}.$$

If IFAIL = 3 or 5 on exit (i.e., MAXIT iterations have been performed but the above condition does not hold), then the elements in A, C, ALPHA and PIGAM may still be good approximations to the maximum likelihood estimates. The user is advised to inspect the elements of G to see whether this is confirmed.

8 Further Comments

8.1 Timing

The number of iterations required in the maximum likelihood search depends upon the number of observed variables, p , and the distance of the user-supplied starting point from the solution. The number of multiplications and divisions performed in an iteration is proportional to p .

8.2 Initial Estimates

The user is strongly advised to use the recommended starting values for the elements of A and C. Divergence may result from user-supplied values even if they are very close to the solution. Divergence may also occur when an item has nearly all its responses at one level.

8.3 Heywood Cases

As in normal factor analysis, Heywood cases can often occur, particularly when p is small and n not very big. To overcome this difficulty the maximum likelihood search routine is terminated when the absolute value of one of the α_{j1} exceeds 10.0. The user has the option of deciding whether to exit from G11SAF (by setting IFAIL = 0 on entry) or to permit G11SAF to proceed onwards as if it had exited normally from the maximum likelihood search routine (setting IFAIL = -1 on entry). The elements in A, C, ALPHA and PIGAM may still be good approximations to the maximum likelihood estimates. The user is advised to inspect the elements G to see whether this is confirmed.

8.4 Goodness of Fit Statistic

When n is not very large compared to s a goodness of fit statistic should not be calculated as many of the expected frequencies will then be less than 5.

8.5 First and Second Order Margins

The observed and expected **percentages** of sample members responding to individual and pairs of items held in the arrays OBS and EXPP on exit can be converted to observed and expected **numbers** by multiplying all elements of these two arrays by $n/100.0$.

9 Example

A program to fit the logit latent variable model to the following data:

Index	Score Pattern	Observed Frequency
1	0000	154
2	1000	11
3	0001	42
4	0100	49
5	1001	2
6	1100	10
7	0101	27
8	0010	84
9	1101	10
10	1010	25
11	0011	75
12	0110	129
13	1011	30
14	1110	50
15	0111	181
16	1111	121
		—
Total		1000

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      G11SAF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
      INTEGER          IPMAX, ICM, ISMAX, NRE, LW, NRX
      PARAMETER       (IPMAX=10,ICM=2*IPMAX,ISMAX=1024,NRE=IPMAX,
+                    LW=4*IPMAX*(IPMAX+16),NRX=ISMAX)
*      .. Local Scalars ..
      real            CGETOL, CHI, RLOGL, SIGLEV
      INTEGER          I, IDF, IFAIL, IP, IPRINT, IS, ISHOW, J, MAXIT,
+                    N, NITER
      LOGICAL          CHISQR, GPROB
*      .. Local Arrays ..
      real            A(IPMAX), ALPHA(IPMAX), C(IPMAX),
+                    CM(ICM,2*IPMAX), EXF(ISMAX), EXPP(NRE,IPMAX),
+                    G(2*IPMAX), OBS(NRE,IPMAX), PIGAM(IPMAX), W(LW),
+                    XL(ISMAX), Y(ISMAX)
      INTEGER          IOB(ISMAX), IRL(ISMAX)
      LOGICAL          X(NRX,IPMAX)
*      .. External Subroutines ..
      EXTERNAL         G11SAF, X04ABF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G11SAF Example Program Results'

```

```

*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) IP, N, IS
      CALL X04ABF(1,NOUT)
      IF (IP.GT.0 .AND. IP.LE.IPMAX .AND. IS.GE.0 .AND. IS.LE.ISMAX)
+      THEN
          DO 20 I = 1, IS
              READ (NIN,*) IRL(I), (X(I,J),J=1,IP)
20      CONTINUE
          GPROB = .FALSE.
          DO 40 I = 1, IP
              A(I) = 0.5e0
              C(I) = 0.0e0
40      CONTINUE
*      ** Set IPRINT $>$ 0 to obtain intermediate output **
          IPRINT = -1
          CGETOL = 0.0001e0
          MAXIT = 1000
          CHISQR = .TRUE.
          ISHOW = 7
          IFAIL = -1
*
          CALL G11SAF(IP,N,GPROB,IS,X,NRX,IRL,A,C,IPRINT,CGETOL,MAXIT,
+              CHISQR,ISHOW,NITER,ALPHA,PIGAM,CM,ICM,G,EXPP,NRE,
+              OBS,EXF,Y,XL,IOB,RLOGL,CHI,IDF,SIGLEV,W,LW,IFAIL)
*
          END IF
          STOP
          END

```

9.2 Program Data

G11SAF Example Program Data

```

4 1000 16
154 F F F F
 11 T F F F
 42 F F F T
 49 F T F F
   2 T F F T
 10 T T F F
 27 F T F T
 84 F F T F
 10 T T F T
 25 T F T F
 75 F F T T
129 F T T F
 30 T F T T
 50 T T T F
181 F T T T
121 T T T T

```

9.3 Program Results

G11SAF Example Program Results

LOG LIKELIHOOD KERNEL ON EXIT = -0.24039E+04

MAXIMUM LIKELIHOOD ESTIMATES OF ITEM PARAMETERS ARE AS FOLLOWS

```

-----
ITEM J      ALPHA(J)      S.E.      ALPHA(J,0)      PI(J)      S.E.
-----
1          1.045          0.148      -1.276          0.218      0.017
2          1.409          0.179         0.424          0.604      0.022
3          2.659          0.525         1.615          0.834      0.036
4          1.122          0.140      -0.062          0.485      0.020
    
```

EXPECTED (AND OBSERVED) PERCENTAGE OF CASES PRODUCING POSITIVE RESPONSES FOR INDIVIDUAL AND PAIRS OF ITEMS

```

-----
ITEM      ITEM
-----
ITEM      1      2      3      4
-----
1          25.9
      (25.9)
2          19.1  57.7
      (19.1)(57.7)
3          22.5  48.0  69.4
      (22.6)(48.1)(69.5)
4          16.4  33.9  40.6  48.8
      (16.3)(33.9)(40.7)(48.8)
    
```

OBSERVED FREQUENCY	EXPECTED FREQUENCY	THETA SCORE	COMPONENT SCORE	RAW SCORE	SCORE PATTERN
154	147.061	-1.273	0.000	0	FFFF
11	13.444	-0.873	1.045	1	TFFF
42	42.420	-0.846	1.122	1	FFFT
49	54.818	-0.747	1.409	1	FTFF
2	5.886	-0.494	2.167	2	TFFT
10	8.410	-0.399	2.455	2	TTFF
27	27.511	-0.374	2.531	2	FTFT
84	92.062	-0.332	2.659	1	FFTF
10	6.237	-0.019	3.577	3	TTFT
25	21.847	0.027	3.705	2	TFTF
75	73.835	0.055	3.781	2	FFTT
129	123.766	0.162	4.069	2	FTTF
30	26.899	0.466	4.826	3	TFTT
50	50.881	0.591	5.114	3	TTTF

181	179.564	0.626	5.190	3	F T T T
121	125.360	1.144	6.236	4	T T T T
-----	-----				
1000	1000.000				

LIKELIHOOD RATIO GOODNESS OF FIT STATISTIC = 9.027
SIGNIFICANCE LEVEL = 0.251
(BASED ON 7 DEGREES OF FREEDOM)

VALUE OF IFAIL PARAMETER ON EXIT FROM G11SAF = 0
